

SINTAXIS DE SQL-92

Introducción:

Se presenta brevemente un resumen de la sintaxis de SQL según el estándar ISO 9075 (SQL-92), dividido en tres partes:

- Lenguaje de Definición de Datos (LDD),
- Lenguaje de Manipulación de Datos (LMD), y
- Lenguaje de Administración de Datos (LAD).

La nomenclatura utilizada es una extensión de la Forma Normal de Backus (BNF) siendo:

- <> los símbolos no terminales del lenguaje
- ::= el operador de definición
- [] elementos opcionales
- { } elementos en una fórmula
- | una alternativa
- ... una repetición

Lenguaje de Definición de Datos

<definición de esquema > ::=
CREATE SCHEMA <cláusula de nombre de esquema>
[<elemento de esquema> ...]

<cláusula de nombre de esquema > ::=
<nombre de esquema>
| AUTHORIZATION <id. de autorización de usuario>
| <nombre de esquema> AUTHORIZATION <id. de autorización de usuario>

<elemento de esquema > ::=
<definición de dominio>
| <definición de tabla>
| <definición de vista>
| <definición de aserción>

<definición de dominio > ::=
CREATE DOMAIN <nombre de dominio> [AS] <tipo de datos>
[<cláusula de defecto>]
[<restricción de dominio>]

<cláusula de defecto > ::=
DEFAULT <opción por defecto>

<opción por defecto> ::=

<literal>
| <función de valor tiempo/fecha>
| USER
| SYSTEM
| USER
| NULL

<restricción de dominio> ::=

[<definición de nombre de restricción>]
<definición de restricción de verificación>
[<atributos de restricción>]

<definición de nombre de restricción> ::=

CONSTRAINT <nombre de restricción>

<definición de restricción de verificación> ::=

CHECK <parent. izq.> <condición de búsqueda> <parent. dcho.>

<atributos de restricción> ::=

<tiempo de verificación de restricción> [[NOT] DEFERRABLE]
| [[NOT] DEFERRABLE] <tiempo de verificación de restricción>

<tiempo de verificación de restricción> ::=

INITIALLY DEFERRED
| INITIALLY IMMEDIATE

<definición de tabla> ::=

CREATE [{GLOBAL|LOCAL} TEMPORARY] TABLE <nombre de tabla>
<lista de elementos de tabla>

<lista de elementos de tabla> ::=

<paréntesis izq.> <elemento de tabla>
[{<coma> <elemento de tabla> } ...] <paréntesis dcho.>

<elemento de tabla> ::=

<definición de columna>
| <definición de restricción de tabla>
<definición de columna> ::=
<nombre de columna> { <tipo de datos> | <nombre de dominio> }
[<cláusula de defecto>]
[<definición de restricción de columna>]

<definición de restricción de columna> ::=

[<definición de nombre de restricción>]
<restricción de columna>
[<atributos de restricción>]

<restricción de columna> ::=
 NOT NULL
 | <especificación de unicidad>
 | <especificación de referencia>
 | <definición de restricción de verificación>

<definición de restricción de tabla> ::=
 [<nombre de definición de restricción>]
 <restricción de tabla>
 [<atributos de restricción>]

<restricción de tabla> ::=
 <definición de restricción de unicidad>
 | <definición de restricción referencial>
 | <definición de restricción de verificación>

<definición de restricción de unicidad> ::=
 <especificación de unicidad>
 <parent. izq.> <lista de columnas únicas> <parent. dcho.>

<especificación de unicidad> ::=
 UNIQUE | PRIMARY KEY

<lista de columnas únicas> ::= <lista de nombre de columnas>

<definición de restricción referencial> ::=
 FOREIGN KEY <parent. izq.> <columnas que ref.> <parent. dcho.>
 <especificación de la referencia>

<especificación de la referencia> ::=
 REFERENCES <columnas y tabla referenciadas>
 [<acción referencial disparada>]

<columnas que ref.> ::= <lista de columnas de referencia>

<columnas y tabla referenciadas> ::=
 <nombre de tabla> [<parent. izq.> <lista de columnas de referencia> <parent. dcho.>]

<lista de columnas de referencia> ::= <lista de nombres de columnas>

<acción referencial disparada> ::=
 <regla de modificación> [<regla de borrado>]
 | <regla de borrado> [regla de modificación]

<regla de modificación> ::= ON UPDATE <acción referencial>

<regla de borrado> ::= ON DELETE <acción referencial>

<acción referencial> ::=
CASCADE
| SET NULL
| SET DEFAULT

<definición de vista> ::=
CREATE VIEW <nombre de tabla>
[<parent. izq.> <lista de columnas de vista> <parent. dcho.>]
AS <expresión de consulta>
[WITH CHECK OPTION]

<lista de columnas de vista> ::= <lista de nombres de columnas>

<definición de aserción> ::=
CREATE ASSERTION <nombre de restricción>
<verificación de aserción> [<atributos de restricción>]

<verificación de aserción> ::=
CHECK <parent. izq.> <condición de búsqueda> <parent. dcho.>

Lenguaje de Manipulación de Datos

<expresión de consulta> ::=
<expresión de consulta de no combinación>
| <tabla combinada>

<expresión de consulta de no combinación> ::=
<término de consulta de no combinación>
| <expresión de consulta> UNION [ALL]
[<especificación de correspondencia>] <término de consulta>
| <expresión de consulta> EXCEPT [ALL]
[<especificación de correspondencia>] <término de consulta>

<término de consulta> ::=
<término de consulta de no combinación>
| <tabla combinada>

<término de consulta de no combinación> ::=
<primario de consulta de no combinación>
| <término de consulta> INTERSECT [ALL]
[<especificación de correspondencia>] <primario de consulta>

<primario de consulta> ::=
<primario de consulta de no combinación>
| <tabla combinada>

<primario de consulta de no combinación> ::=
<tabla simple>
| <parent. izq.> <expresión de combinación de no consulta> <parent. dcho.>

<cláusula as> ::= [AS] <nombre de columna>

<cuantificador de conjunto> ::= DISTINCT | ALL

<expresión de tabla> ::=
 <cláusula from>
 [<cláusula where>]
 [<cláusula group by>]
 [<cláusula having>]

<cláusula from> ::= FROM <referencia a tabla> [{ <coma> <referencia a tabla> } ...]

<referencia a tabla> ::=
 <nombre de tabla> [[AS] <nombre de correlación>
 [<parent. izq.> <lista de columnas derivadas> <parent. dcho.]]
 | <tabla derivada> [AS] <nombre de correlación>
 [<parent. izq.> <lista de columnas derivadas> <parent. dcho.]
 | <tabla combinada>

<tabla derivada> ::= <subconsulta de tabla>

<lista de columnas derivadas> ::= <lista de nombres de columnas>

<lista de nombres de columnas> ::=
 <nombre de columna> [{ <coma> <nombre de columna> } ...]

<nombre de correlación> ::= <identificador>

<calificador> ::=
 <nombre de tabla>
 | <nombre de correlación>

<tabla combinada> ::=
 <combinación cruzada>
 | <combinación calificada>
 | <parent. izq.> <tabla combinada> <parent. dcho.>

<combinación cruzada> ::=
 <referencia a tabla> CROSS JOIN <referencia a tabla>

<combinación calificada> ::=
 <referencia a tabla> [NATURAL] [<tipo de combinación>] JOIN <referencia a tabla>
 [<especificación de combinación>]

<especificación de combinación> ::=
 <condición de combinación>
 | <combinación de columnas nominadas>

<condición de combinación> ::= ON <condición de búsqueda>

<combinación de columnas nominadas> ::=
 USING <parent. izq.> <lista de columnas de combinación> <parent. dcho.>

<tipo de combinación> ::=
 INNER
 | <tipo de combinación externa> [OUTER]
 | UNION

<tipo de combinación externa> ::=
 LEFT
 | RIGHT
 | FULL

<lista de columnas de combinación> ::= <lista de nombres de columnas>

<cláusula where> ::= WHERE <condición de búsqueda>

<cláusula group by> ::=
 GROUP BY <lista de referencias a columna de agrupamiento>

<lista de referencias a columna de agrupamiento> ::=
 <referencia a columna de agrupamiento>[{ <coma> <referencia a columnas de agrupamiento> } ...]

<referencia a columna de agrupamiento> ::=
 <referencia a columnas>

<referencias a columnas> ::=
 [<calificador> <punto>] <nombre de columna>

<cláusula having> ::= HAVING <condición de búsqueda>

<condición de búsqueda> ::=
 <término booleano>
 | <condición de búsqueda> OR <término booleano>

<término booleano> ::=
 <factor booleano>
 | <factor booleano> AND <factor booleano>

<factor booleano> ::=
 [NOT] <test booleano>

<test booleano> ::=
 <primario booleano> [IS [NOT] <valor lógico>]

<valor lógico> ::=
 TRUE
 | FALSE
 | UNKNOWN

<primario booleano> ::=
 <predicado>
 | <parent. izq.> <condición de búsqueda> <parent. dcho.>

<predicado> ::=
 <predicado de comparación>
 | <predicado entre>
 | <predicado en>
 | <predicado parece>
 | <predicado de nulidad>
 | <predicado de comparación cuantificada>
 | <predicado existe>
 | <predicado de unicidad>
 | <predicado de concordancia>
 | <predicado de solapamiento>

<predicado de comparación> ::=
 <constructor de valor de fila> <operación de comparación> <constructor de valor de fila>

<predicado entre> ::=
 <constructor de valor de fila> ::= [NOT] BETWEEN
 <constructor de valor de fila> AND <constructor de valor de fila>

<predicado en> ::=
 <constructor de valor de fila>
 [NOT] IN <valor de predicado en>

<valor de predicado en> ::=
 <subconsulta de tabla>
 | <parent. izq.> <lista de valores en> <parent. dcho.>

<lista de valores en> ::=
 <expresión de valor> { <coma> <expresión de valor> } ...

<predicado parece> ::=
 valor de concordancia> [NOT] LIKE <patrón>

<valor de concordancia> ::= <expresión de valores caracter>

<patrón> ::= <expresión de valores caracter>

<predicado de nulidad> ::=
 constructor de valor de fila> IS [NOT] NULL

<predicado de comparación cuantificada> ::=
 constructor de valor de fila> <operación de comparación>
 <cuantificador> <subconsulta de tabla>

<cuantificador> ::= <todos> | <algunos>

<todos> ::= ALL

<algunos> ::= SOME | ANY

<predicado existe> ::= EXISTS <subconsulta de tabla>

<predicado de unicidad> ::= UNIQUE <subconsulta de tabla>

<predicado de concordancia> ::=
 <constructor de valor de fila>MATCH[UNIQUE][PARTIAL | FULL]<subconsulta de
tabla>

<predicado de solapamiento> ::=
 <constructor de valor de fila 1> OVERLAPS <constructor de valor de fila 2>

<constructor de valor de fila 1> ::= <constructor de valor de fila>

<constructor de valor de fila 2> ::= <constructor de valor de fila>

Otras sentencias de manipulación de datos son las siguientes:

<sentencia de borrado: con búsqueda> ::=
 DELETE FROM <nombre de tabla>
 [WHERE <condición de búsqueda>]

<sentencia de inserción> ::=
 INSERT INTO <nombre de tabla>
 <fuente y columnas de inserción>

<fuente y columnas de inserción> ::=
 [<parent. izq.> <lista de columnas de inserción> <parent. dcho.>] <expresión de
consulta>
 | DEFAULT VALUES

<lista de columnas de inserción> ::=
 <lista de nombres de columnas>

<sentencia de modificación: con búsqueda> ::=
 UPDATE <nombre de tabla>
 SET <lista de cláusula de conjunto>
 [WHERE <condición de búsqueda>]

<lista de cláusula de conjunto> ::=
 <cláusula de conjunto> [{<coma> <cláusula de conjunto>}...]

<cláusula de conjunto> ::=
 <nombre de columna> <operador igual> <fuente de modificación>

<fuente de modificación> ::=
 <expresión de valor>
 | <especificación de nulos>
 | DEFAULT

Lenguaje de Administración de Datos

<sentencia de concesión> ::=
 GRANT <privilegios> ON <nombre de objeto>
 TO <concedido> [{ <coma> <concedido> } ...]
 [WITH GRANT OPTION]

<nombre de objeto> ::=
 [TABLE] <nombre de tabla>
 | DOMAIN <nombre de dominio>

<sentencia de revocación> ::=
 REVOKE [GRANT OPTION FOR] <privilegios>
 ON <nombre de objeto>
 FROM <concedido> [{ <coma> <concedido> } ...] <comportamiento al borrar>

<privilegios> ::=
 ALL PRIVILEGES
 | <lista de acciones>
 <lista de acciones> ::= <acción> [{ <coma> <acción> } ...]

<acción> ::=
 SELECT
 | DELETE
 | INSERT [<parent. izq.> <lista de nombres de columnas> <parent. dcho.>]
 | UPDATE [<parent. izq.> <lista de nombres de columnas> <parent. dcho.>]
 | REFERENCES [<parent.izq.> <lista de nombresdecolumnas> <parent. dcho.>]
 | USAGE

<concedido> ::=
 PUBLIC
 | <id. de autorización de usuario>

Anotaciones:

1.- Relativas al <tipo de datos>

a) En SQL-92, además de los tipos de datos del SQL-89 (INTEGER, SMALLINT, CHARACTER, DECIMAL, NUMERIC, REAL, FLOAT y DOUBLE PRECISION) también se admiten los siguientes: CHARACTER VARYING, DATE, TIME, BIT, TIMESTAMP, INTERVAL y BIT VARYING.

b) Los tipos BIT y BIT VARYING contienen bits pero no son interpretados por el SGBD; INTERVAL expresa la diferencia entre fechas u horas; y TIMESTAMP contiene año, mes, día, hora, minutos, segundos y, opcionalmente, fracciones de segundo.

2.- Relativas a <expresión de valor>

a) Una <expresión de valor> es cualquier expresión que dé como resultado un valor numérico, carácter, hora, fecha o intervalo. Puede ser tan simple como un dígito o tan complejo como una subconsulta.

3.- Relativas a la <definición de restricción referencial>

a) Si <columnas y tabla referenciada> especifica una <lista de columnas de referencia>, ésta debe ser idéntica a una <lista de columnas únicas> en una <definición de restricción de unicidad> de la tabla referenciada. Por tanto, se puede establecer una restricción referencial con conjuntos de columnas únicas sin que sean necesariamente la clave primaria. En caso de no especificar la <lista de columnas de referencia> se entenderá que se referencia a la clave primaria.

b) En caso de no especificar la <acción referencial disparada> se considera que ésta es restringida.

c) Las propuestas del estándar son más completas, incluyendo otros aspectos de las restricciones de integridad referencial como el tipo de concordancia (si todos los valores de la columnas que referencian pueden ser nulos o no nulos o ambas cosas) y el modo de restricción (diferida, si se verifica al finalizar la transacción o inmediata, si se efectúa al finalizar cada sentencia).

4.- Relativas a la <operación de comparación>

a) Las operaciones de comparación son las usuales: =, >, <, >=, <=, <> .

5.- Relativas a la dinámica

a) Hay que destacar que además de las operaciones presentadas existen otras relacionadas con el manejo de los cursores en SQL embebido.